
Met joystick en cursortoets We zetten de besturing met joystick en cursortoets nog 
eens netjes op een rij. Daarnaast geven we een programma waarmee je beide kunt 
uitlezen en die zelfs een tweede vuurknop op een toetsenbord biedt. 
 
Het MSX BIOS heeft de routines GTSTCK (00D5h) en GTTRIG (00D8h) om 
respectievelijk de cursortoetsen of joystick en de vuurknoppen uit te kunnen lezen. 
Deze routines zijn echter nogal traag en gebruiken veel registers. Bovendien kun je op 
het toetsenbord voor de vuurknop alleen de spatiebalk uitlezen en geen andere toets 
die als tweede trigger kan dienen. Je kunt de routines dus niet universeel gebruiken 
om zowel joystick als toetsenbord te ondersteunen. 
Het programmaatje GETJOY.ASM ruimt deze problemen uit de weg. Het biedt naast 
een hogere uitvoersnelheid ook nog eens een tweede vuurknop voor 
het toetsenbord: |SHIFT|. Een bijkomend voordeel is dat GETJOY gebruik maakt van 
een minimum aan registers. 
 



 

 

 



Matrix 
 
Zoals je waarschijnlijk wel weet, maken MSX computers gebruik van zogenaamde 
matrix-toetsenborden die per rij zijn uit te lezen. In de tabel op de pagina hiernaast is 
de indeling van deze rijen vermeld. Het MSX BIOS scant elke inter-rupt de hele 
matrix. Bij de meeste in Nederland voorkomende MSX'en is dat standaard vijftig keer 
per seconde: de interrupt-frequentie is 50 Hz. Op sommige computers—vooral de 
Japanse— bedraagt deze frequentie echter 60 Hz; de periode tussen twee scans van 
het toetsenbord is dan slechts 1/60 seconde. De gebruikte frequentie wordt echter 
bepaald door de gebruiker, die de VDP, die alle tijd-interrupts genereert, op dit gebied 
zelf kan instellen. 
Het resultaat van de scan wordt op de RAM-adressen FBE5h tot en met FBEFh gezet. 
Door de nieuwe inhoud van de matrix te vergelijken met de oude, kan het BIOS 
nieuw ingedrukte toetsen detecteren en deze in de keyboardbuffer zetten. De 
toetsenbordhardware maakt gebruik van zogenaamde negative logic. Voor de meeste 
mensen is het normaal de uitgangstoestand voor te stellen met een O en een l te 
gebruiken voor een verandering. In het geval van de toetsenbordscan wordt de 
verandering, een ingedrukte toets, juist door een 0 aangegeven en niet door een 1. De 
signalen zijn dus omgedraaid; vandaar de term negative logic. 
 
Lay-out toetsenbordmatrix 
 
adres rij kolombits        
  7 6 5 4 3 2 1 0 
FBE5h 0 7 6 5 4 3 2 1 0 
FBE6h 1 ; ] [ A = - 9 8 
FBE6h 2 B A ACCENTEN /  f ' ' 
FBE6h 3 J 1 H G F E D C 
FBE9H 4 R Q P O N M L K 
FBEAh 5 Z Y X W V U T S 
FBEBh 6 F3 F2 F1 CODE CAPS GRAPH CTRL SHIFT 
FBECh 7 RETURN SELECT BS STOP TAB ESC F5 F4 
FBEDh 8 RIGHT DOWN. UP LEFT DEL INS CLR/HOME SPACE 
FBEEh* 9 4 3 2 1 0 / + * 
FBEFh* 10 . , - 9 8 7 6 5 
 
Uitlezen toetsenbord 
 
In een MSX zitten een aantal hardware-onderdelen die door de CPU, Z80 of R800 
aangesproken moeten worden. Dit gebeurt door een PPI—Programmable Pe-ripheral 
Interface—die eigenlijk niets anders is dan een doorgeefluik, in dit geval de 8255 PPI. 
De 8255 hier heeft niets met de NMS 8255 van Philips te maken. Dit kleine stukje 
hardware dient in feite als buffer tussen de CPU en de rest van de hardware. De 8255 
PPI is verbonden aan de CPU door middel van vier poorten, waarvan er drie werkelijk 
dienen voor het doorgeven van data en de laatste de mogelijkheid biedt om de 8255 
PPI zelf in te stellen. Het linker kader op de vorige pagina geeft alle details. Je kunt 
een kolom dus lezen door eerst het kolomnummer in KO to K3 van PPI-poort C te 
zetten en vervolgens PPI-poort B te lezen. Het BIOS dumpt de verkregen waarde 
gewoon in de tabel, maar je kunt er zelf natuurlijk bepaalde bitjes uit isoleren en er 
wat anders mee doen. Je kunt nu dus ook reageren op toetsen die normaal niet in de 
keyboardbuffer worden gezet, zoals bij voorbeeld: |SHIFT|, [CTRL], |CODE|. Als je 
trouwens steeds dezelfde kolom wilt uitlezen hoef je PPI-poort C niet steeds opnieuw 
in te stellen. 



 
Joystickscan 
 
Voor het lezen van de joysticks geldt zo ongeveer hetzelfde verhaal: de hardware—de 
joysticks—wordt via een PPI aan de CPU gekoppeld. Het bijzondere is echter dat 
hiervoor de PSG wordt gebruikt! De PSG heeft namelijk twee acht-bits datapoorten 
over. Zij worden A en B genoemd. Dit was natuurlijk     
ontzettend handig voor de ontwerpers van de MSX want op deze manier hoefden ze 
geen extra hardware te gebruiken. De 8910 PSG is met drie poorten verbonden aan de 
CPU waarmee het aangestuurd kan worden. Zie hiervoor het rechter kader op de 
vorige pagina. De PSG bevat zestien registers waarvan er veertien gebruikt worden 
om de drie geluidskanelen mee aan te sturen. De overige twee hebben een andere 
functie gekregen. Zie het kleine kader op de eerste pagina van dit artikel. 
Om een bepaald PSG-register aan te spreken moet je dus altijd zorgen dat het 
adresregister naar dat register wijst. Je hoeft ook hier niet steeds het adresregister 
steeds opnieuw in te stellen als je het zelfde register blijft aanspreken. Om een 
joystick te lezen moet je eerst aangeven welke joystick met behulp van bit 6 van PSG 
register 15. Daarna kun je de 'status' verkrijgen door PSG register 14 te lezen. Dit 
register kan ook meerdere malen uitgelezen worden zonder register 14 opnieuw in te 
stellen. 
 
De ml-routines behoeven denk ik geen verder uitleg. Het commentaar bij de assembly 
source zal voor de meesten wel voldoende zijn. Al moet worden toegegeven dat de 
routines door een kleine optimalisatie wel een beetje onleesbaarder zijn geworden. Dit 
is natuurlijk gedaan om zo de uitvoersnel-heid van dit programma te verhogen. Zowel 
treft u mijn originele source aan en de hiernaast geplaatste voor publicatie aangepaste 
versie. Veel problemen zal dit echter niet geven. 
 
 



 
 
 ORG &HA000 
 
;Get Stick and Triggers (GTSTTR) 
;In: A = Stick# : 0 = keyboard 
;                 1 = joystick 1 
;                 2 = joystick 2 
;Out: A = Steering direction (0-8) 
;     C = trigger A (0 = not pressed, 1 = pressed) 
;     B = trigger B (0 = not pressed, 1 = pressed) 
;Uses: F,H,L 
;Note: PSGREG and PPIPRC won't be affected 
 
PSGREG: EQU &HA0 ;PSG REGister pointer port# 
PSGWDT: EQU &HA1 ;PSG Write DaTa into register port# 
PSGRDT: EQU &HA2 ;PSG Read DaTa from register port# 
 
PPIPRB: EQU &HA9 ;PPI PoRt B 
PPIPRC: EQU &HAA ;PPI PoRt C 
 
GTSTTR:   ; Goede Tijden Slechte Tijden Televisie Rotzooi? :) 
 
 DI 
 
;which stick? 
 AND A 
 JP Z,JUMP 
 
;get stick 1 or 2 
 CPL 
 AND &B00000001 ;reposition bit#0 
 RRCA 
 RRCA 
 LD B,A 
 LD C,PSGREG ;store current contents in H 
 IN H,(C) 
 LD A,15 ;PSG register #15 
 OUT (C),A 
 IN A,(PSGRDT) ;set active joystick 
 AND &B10111111 
 OR B 
 OUT (PSGWDT),A 
 LD A,14 ;PSG regsister #14 
 OUT (C),A ;get joystick status 
 IN A,(PSGRDT) ;A = steering direction + triggers 
 OUT (C),H ;reset register pointer 
 
;calculate STEERING DIRECTION 
 LD L,A ;keep a copy in L 
 AND &B00001111 
 LD C,A 
 LD B,0 
 LD A,L ;shift copy back to A 
 LD HL,DIRTBJ 
 ADD HL,BC ;HL points at Steering Direction 
 
;calculate TRIGGER B 
 CPL ;(negative logic) 
 AND &B00110000 ;use copy in A 
 ADD A,A 
 ADD A,A 
 ADD A,A ;Cy = trigger B 
 RL B 
@LIST-VOET = GET_JOY.ASM <F129M><188><F255D> 
@LIST-KOP = <F129M><188><F255D> ML-LISTING 
;calculate TRIGGER A 
 RLCA 
 LD C,A 
 LD A,(HL) ;A = Steering Direction 
 
 EI 
 
 RET 
 
;DIRection TaBle for Joy-sticks 



DIRTBJ: DEFB 0,0,0,0,0,4,2,3,0,6,8,7,0,5,1,0 
 
;get stick 0 
JUMP: 
 IN A,(PPIPRC) ;store current contents in H 
 LD H,A 
 AND &B11110000 ;reset column bits only 
 LD L,A 
 
;calculate TRIGGER B 
 OR 6 ;read column 6 
 OUT (PPIPRC),A 
 IN A,(PPIPRB) 
 LD C,&B00000001 ;C= bitmask, 
  ;(sorry about that C64 logo) :) 
 AND C 
 XOR C ;(negative logic) 
 LD B,A 
 
;calculate TRIGGER A 
 LD A,8 ;read column 8 
 OR L 
 OUT (PPIPRC),A 
 IN A,(PPIPRB) 
 LD L,A ;keep a copy in L 
 AND C 
 XOR C ;(negative logic) 
 LD C,A 
 PUSH BC ;save triggers 
 
;calculate STEERING DIRECTION 
 LD A,&B11110000 ;use copy in L 
 AND L 
 RRCA 
 RRCA 
 RRCA 
 RRCA 
 LD C,A 
 LD B,0 
 LD A,H ;shift original contents to A 
 LD HL,DIRTBC 
 ADD HL,BC ;HL points at Steering Direction 
 OUT (PPIPRC),A ;reset register with original value 
 LD A,(HL) ;wop in accum. 
 POP BC 
 
 EI 
 
 RET 
 
;DIRection TaBle for Cursor 
DIRTBC: DEFB 0,3,5,4,1,2,0,3,7,0,6,5,8,1,7,0 

